

---

Subject: Re: How do you access a field in a struct?  
Posted by [m218003](#) on Wed, 15 Dec 1999 08:00:00 GMT  
[View Forum Message](#) <> [Reply to Message](#)

---

In article <38573DF7.39E554F9@twins.dna.fi>, Karri Kaksonen <karri@twins.dna.fi> writes:

```
> Hi,
>
> I have a small problem with accessing my structs.
>
> The complete program (that does not work) looks like:
> pro beer
>   a = { $
>     beer: 1, $
>     icecream: 2 $
>   }
>
>   b = { $
>     milk: 0.60, $
>     beer: 1.25, $
>     icecream: 0.75, $
>     tea: 0.50 $
>   }
>
>   totalsum = 0.0
>   selection = tag_names(a)
>   for i=0,n_tags(a)-1 do totalsum=totalsum + a.(i) * b.(selection)
>   print, totalsum
> end
>
> When I run this program I get:
> IDL> beer
> % Type conversion error: Unable to convert given STRING to Long.
> % Detected at: BEER          16 C:\RSI\IDL52\mri\beer.pro
> % Type conversion error: Unable to convert given STRING to Long.
> % Detected at: BEER          16 C:\RSI\IDL52\mri\beer.pro
>    1.80000
>
> The funny thing is that it obviously replaced "beer" and "icecream"
> with "milk" as it could not understand that b.('beer') is the same
> as b.beer
>
```

Sure enough, IDL complains. Insert a  
HELP, selection  
statement, and you'll see that selection is of type string.  
The "tag selector" b.(i) requires an integer, however. The solution to your  
problem is the WHERE function. The core statement is

```
bind = WHERE(tag_names(b) eq selection[i])
```

However, for reasons of efficiency and user-tampering-protection, the code would rather look like:

```
totalsum = 0.0
selection = tag_names(a)
bnames = tag_names(b)
for i=0,n_tags(a)-1 do begin
  bind = WHERE(bnames eq selection[i],count)
  if (count eq 1) then begin
    totalsum=totalsum + a.(i) * b.(bind[0])
  endif else begin
    print,'No matching tag in b for ',selection[i]
  endelse
endfor
print, totalsum
```

BTW: the answer is 2.75 ;-)

Cheers,  
Martin

--

```
[[ Dr. Martin Schultz  Max-Planck-Institut fuer Meteorologie  [[
[[          Bundesstr. 55, 20146 Hamburg          [[
[[          phone: +49 40 41173-308          [[
[[          fax: +49 40 41173-298          [[
[[ martin.schultz@dkrz.de          [[
[[          [[          [[          [[          [[          [[          [[
```

---